Funkce a její vlastnosti

Zadávání funkcí

Nejdříve se podíváme na zadávání funkcí pomocí jedné rovnosti, později na zadávání funkcí po částech.

Je duležité znát rozdíl mezi zadáváním hodnot pomocí = a := a mezi proměnnou x a x_ . Druhá uvedená rovnost znamená pozdržené použití (i když je za x předem dána nějaká hodnota, na definici to nemá vliv, kdežto u první rovnosti ano). Podtržítko za proměnnou x znamená, že se za x můře dosazovat jakýkoli vzorec s jednou proměnnou.

Rozdíly lze pochopit z následujících příkladů. Nyní použijeme definici f[x]=. Budeme definovat funkci a pak dosazovat proměnné - tím ale hodnoty funkce nedostaneme.

```
Clear[x]; Clear[h]; h[x] = 2^x
```

2^x

```
x = 3
```

3

```
h[x]
```

$h(3)$

```
h[3]
```

$h(3)$

```
h[x] /. x → 4
```

$h(4)$

Nyní máme dánu hodnotu proměnné x=3 a budeme znovu definovat funkci. Dostaneme funkční hodnotu v dané proměnné, ale jen pro tuto hodnotu x.

```
Clear[h]; h[x] = 2^x
```

8

```
h[4]
```

$h(4)$

```
x = 5; h[x]
```

$h(5)$

```
Clear[g]; g[x] = 3^x
```

243

Použijeme nyní pro definici f[x]:=. Výsledky se nezmění.

```
Clear[g]; g[x] := 3^x
```

```
g[4]
```

$g(4)$

```
g[x]
```

243

Vrátíme se k první jednoduché rovnosti, ale použijeme v definici x_ . Ze začátku se zdá, že vše pracuje dobře.

```
Clear[f]; Clear[x]; f[x_] = x^2
```

x^2

```
f[x]
```

x^2

```
f[7]
```

49

```
x = 5; f[x]
```

25

```
f[4]
```

16

Pokud ale před definicí funkce máme hodnotu proměnné zadanou, definice dobře nepracuje.

```
Clear[f]; f[x_] = x^2
```

25

```
f[3]
```

25

Teprve poslední možnost definice funkce f[x_]:= je v pořádku.

```
Clear[f]; f[x_] := x^2
```

```
f[x]
```

25

```
f[3]
```

9

```
x = 6; f[x]
```

36

```
f[y]
```

y^2

```
f[2]
```

4

Definiční obor a obor hodnot funkce

Určovat definiční obor funkce (pokud není zadán) se dá programem *Mathematica* jen ve speciálních případech. Ukážeme, kdy to jde i kdy to nejde nebo kdy dostaneme podivný výsledek. Základním příkazem je Reduce, občas pomůže FindInstance (viz průběh funkce) nebo nakreslení grafu pomocí příkazu Plot, který bude probrán později (graf se nenakreslí mimo definiční obor). Je nutné přidat Reals na konec příkazu Reduce, protože pracujeme jen s reálnými funkcemi reálné proměnné (je možné dát např. Integers nebo Complexes). Příkaz Reduce může určit i obor hodnot.
Z prvních čtyř příkladů na odmocninu je jasně vidět, jaký má na výsledek vliv výskyt a pořadí proměnných.

```
Clear[x, y]; Reduce[y == Sqrt[x], y, Reals]
```

$x \geq 0 \bigwedge y = \sqrt{x}$

```
Reduce[y == Sqrt[x], x, Reals]
```

$y \geq 0 \bigwedge x = y^2$

```
Reduce[y == Sqrt[x], y, x, Reals]
```

$y \geq 0$

```
Reduce[y == Sqrt[x], x, y, Reals]
```

$x \geq 0$

Další různé příklady.

```
Reduce[y == Abs[x], y, x, Reals]
```

$y \geq 0$

```
Reduce[y == (x^2 + 1) / x, x, y, Reals]
```

$x < 0 \bigvee x > 0$

```
Clear[x, y]; Reduce[y == ArcSin[x], x, y, Reals]
```

$-1 \leq x \leq 1$

```
Clear[x, y]; Reduce[y == ArcSin[x], y, x, Reals]
```

$-\frac{\pi}{2} \leq y \leq \frac{\pi}{2}$

```
Reduce[y == Tan[x], x, y, Reals]
```

$c_1 \in \mathbb{Z} \bigwedge \left(\frac{1}{2}(4\pi c_1 + \pi) < x < \frac{1}{2}(4\pi c_1 + 3\pi) \bigvee \frac{1}{2}(4\pi c_1 - \pi) < x < \frac{1}{2}(4\pi c_1 + \pi)\right)$

```
Reduce[y == Sqrt[Sin[x] + 1], x, y, Reals]
```

$$\left(c_1 \in \mathbb{Z} \bigwedge \left(x = 2\pi c_1 - \frac{\pi}{2} \bigvee x = 2\pi c_1 + \frac{3\pi}{2}\right)\right) \bigvee$$
$$\left(c_1 \in \mathbb{Z} \bigwedge \frac{1}{2}(4\pi c_1 - \pi) < x < \frac{1}{2}(4\pi c_1 + 3\pi)\right)$$

```
Simplify[%]
```

$c_1 \in \mathbb{Z} \bigwedge (2x + \pi = 4\pi c_1 \bigvee 2x = \pi(4c_1 + 3) \bigvee (2x + \pi > 4\pi c_1 \bigwedge 2x < \pi(4c_1 + 3)))$

```
Clear[x]; Clear[y];
Reduce[y == Sqrt[x^2 - 1], x, y, Reals]
```

$x \leq -1 \bigvee x \geq 1$

```
Reduce[y == 1 / (x^2 - 1), x, y, Reals]
```

$x < -1 \bigvee -1 < x < 1 \bigvee x > 1$

```
Reduce[y == a^x, x, y, Reals]
```

$$(a = 0 \wedge x > 0) \bigvee \left(\left(\frac{x}{2} \middle| c_1\right) \in \mathbb{Z} \bigwedge a < 0 \bigwedge x = c_1\right) \bigvee \left(\left(\frac{x+1}{2} \middle| c_1\right) \in \mathbb{Z} \bigwedge a < 0 \bigwedge x = c_1\right) \bigvee a > 0$$

```
Reduce[y == 1 / Sin[x * Pi], x, y, Reals]
```

$$c_1 \in \mathbb{Z} \wedge (2\,c_1 - 1 < x < 2\,c_1 \vee 2\,c_1 < x < 2\,c_1 + 1)$$

Problémy opět nastanou s lichými odmocninami. Podle předchozího příkladu nejsou definovány pro záporná čísla - ověříme pomoci Reduce a Plot.

```
Reduce[y == x^(2 / 3), x, y, Reals]
```

$$x \geq 0$$

```
Plot[x^(2 / 3), {x, -2, 2}]
```



Když se nahraje balíček RealOnly, Reduce opět trvá na nezáporných číslech, teprve Plot ukáže spravný definiční obor. Po ukázce ale balíček odisntalujeme, protože by ovlivnil další výpočty.

```
<< Miscellaneous`RealOnly`
```

— *General::obspkg :*

Miscellaneous`RealOnly` is now obsolete. The legacy version being loaded may conflict with current Mathematica functionality. See the Compatibility Guide for updating information. >>

```
Reduce[y == x^(2/3), x, y, Reals]
```

$x \geq 0$

```
Plot[x^(2/3), {x, -2, 2}]
```



Použijeme definici obecné mocniny a teprve teď je výsledek správný.

```
mocnina[x_, a_] :=
 If[Element[x, Reals] && x >= 0, x^a,
   If[Element[x, Reals] && x < 0 && Element[a, Rationals] &&
     (EvenQ[Numerator[a]] ||
       OddQ[Numerator[a] * Denominator[a]]),
    (2 * Boole[EvenQ[Numerator[a]]] - 1) * (-x)^a, "Nejde"]] //
  N
```

```
Reduce[y == mocnina[x, 2/3], x, y, Reals]
```

— *Reduce::ratnz :*

Reduce was unable to solve the system with inexact coefficients. The answer was obtained by solving a corresponding exact system and numericizing the result. ≫

$x \geq 0 \lor x < 0$

```
Plot[mocnina[x, 2 / 3], {x, -2, 2}]
```



Hodnoty funkce ve více bodech

Funkci tedy vždy zadáváme pomocí f[x_]:=. Její hodnoty pak získáme dosazením příslušných hodnot proměnných. Chceme-li získat najednou hodnoty ve více bodech (bude se to hodit v hledání extrému funkce), máme následující 3 možnosti.

```
Clear[f]; f[x_] := (x^5 - 2 * x^2 + 6 * x - 20) / (x - 1)^2
```

```
body = {-2, 0, 4}
```

$\{-2, 0, 4\}$

```
Map[f, body]
```

$\left\{-8, -20, \frac{332}{3}\right\}$

```
Table[f[k], {k, -2, 4, 2}]
```

$\left\{-8, -20, 16, \frac{332}{3}\right\}$

```
f[a] /. {{a → -2}, {a → 0}, {a → 4}}
```

$\left\{-8, -20, \frac{332}{3}\right\}$

Vlastnosti funkce

Zjišťování vlastností funkce se v Mathematice omezuje ponejvíce na zjišťování, zda je funkce lichá,sudá či periodická nebo zda jsou dvě dané funkce navzájem inverzní.

```
Sin[x] == Sin[-x]
```

$\sin(x) = -\sin(x)$

```
Sin[x] == -Sin[-x]
```

True

```
Cos[x] == Cos[-x]
```

True

```
f[x_] := (x^3 - x) / (x^4 + 8 * x^2)
```

```
f[x] == -f[-x]
```

$$\frac{x^3 - x}{x^4 + 8x^2} = -\frac{x - x^3}{x^4 + 8x^2}$$

```
Tan[x + 3 * Pi] == Tan[x]
```

True

```
Sin[x + Pi] == -Sin[x]
```

True

```
Tan[ArcTan[x]] == x
```

True

```
ArcTan[Tan[x]] == x
```

$\tan^{-1}(\tan(x)) = x$

To je v pořádku. Odpověď by ale měla být True na intervalu (-π/2,π/2):

```
ArcTan[Tan[x]] == x /; -Pi / 2 < x < Pi / 2
```

$\tan^{-1}(\tan(x)) = x \,/;\, -\frac{\pi}{2} < x < \frac{\pi}{2}$

```
Exp[Log[x]] == x
```

True

```
Log[Exp[x]] == x
```

$\log(e^x) = x$

```
InverseFunction[Tan]
```

ArcTan

```
InverseFunction[ArcTan]
```

Tan